



## ITM etalon modelling for Advanced Virgo Plus

VIR-0062A-21

Andreas Freise, Michal Was

*Issue:* 1

*Date:* January 25, 2021

## Contents

<b>1</b>	<b>Introduction</b>	<b>1</b>
<b>2</b>	<b>Setting up the model</b>	<b>2</b>
<b>3</b>	<b>Re-tuning the interferometer operating point using Finesse locks</b>	<b>2</b>
<b>4</b>	<b>Exploring the parameter space, tuning both etalons</b>	<b>3</b>
4.1	Quality function . . . . .	3
4.2	Matched, plane-wave model (maxtem off) . . . . .	4
4.3	Control requirements . . . . .	4
4.4	Mis-matched model (maxtem 2) . . . . .	9
<b>5</b>	<b>Appendix 1: the Finesse file</b>	<b>11</b>

## 1 Introduction

Virgo uses input test masses (ITMs) with plane-parallel surfaces which create an etalon effect; this can be used for in-situ balancing of the arm cavities in Advanced Virgo [1]. For some recent information on the plans for the etalon and its implementation, see for example the Virgo presentations [3, 4].

This note summarizes recent FINESSE [5, 6] simulations to confirm that a) the CARM to DARM coupling can be improved through the etalon tuning, b) that the etalon does not cause any unexpected excess noise, and a) that the expected control accuracy of the etalon temperature is sufficient.




Figure 1: This graph shows CARM to DARM transfer function. The two traces compare of the results obtained earlier with an Optickle model and the results with the current FINESSE model. The FINESSE model was not specifically tuned to match the Optickle results, but uses the best available parameters for the detector at the time of writing.

## 2 Setting up the model

The FINESSE model used in this work is based on previous files developed by Michele Vallisneri and others for ISC and OSD tasks. The file was updated and Pykat scripts were used to prepare the ‘locks’, and later also a mode-mismatched input file. The input files is attached in sectionx 5 and the script files are accessible in a git repository [2].

A simple validation check has been done to confirm that the quantum-noise limited sensitivity of the model is similar to the expected O4 curve. In addition, figure 1 shows the comparison of the CARM to DARM transfer function calculated earlier with an Optickle model by Michal Was with the default setup in the new FINESSE model. Despite the fact that the FINESSE model was not tuned to match the older Optickle parameters, the transfer functions are remarkably similar and serve as a simple validation of the FINESSE input file.

As a figure of merit for the arm cavity balancing we use the CARM to DARM transfer function, computed as the ration between the transfer function DARM→B1 and CARM→B1 [7]. We will denote this transfer function as  $C_{N,W}(f)$  in the following, with  $N$  and  $W$  the tunings of the North and West etalons in degrees respectively.




Figure 2: The top graph shows CARM to DARM transfer functions for a number of tunings of the etalon in the west input test mass (WI) without adjusting the interferometer operating point. The lower graph shows the same calculations but now using FINESSE’s build-in locks to keep the interferometer always on it’s operating point.

## 3 Re-tuning the interferometer operating point using Finesse locks

When modelling the noise coupling it is important to take into account the effect of the etalon tuning on the other degrees of freedom. For example, different tunings of the etalon in the WI mirror would change the phase

of the light in the west arm and thus the dark fringe offset. In the FINESSE simulations we can use so-called ‘locks’ based on the current control design for the five main interferometer degrees of freedom (DARM, CARM, MICH, PRCL, SRCL). These locks automatically re-tune all degrees of freedom for any etalon change such that all error signals remain zero. Figure 2 shows a comparison of the CARM to DARM transfer functions computed with and without such locks. It is clear to see that without the locks the shape of the transfer function is often dominated by the detuning of the interferometer, rather than the wanted adjustment of the arm cavity finesse and reflectivity. All further simulation results shown below were computed with the lock enabled.

## 4 Exploring the parameter space, tuning both etalons

In the examples above we only tuned the WI etalon while keeping the NI etalon at an (arbitrary) tuning of 0. More realistically we need to consider the full phase space defined as the tuning of both etalons (0 to 180 degrees in FINESSE tunings). The disadvantage of using locks is that they can slow down the simulation significantly. Therefore the majority of results below are computed using a plane-wave model (maxtem off) which are faster, and we only check the impact of mode-mismatch in a model with higher-order modes in a dedicated simulation at the end.




Figure 3: Exploring the full parameter space of the etalon tuning. The colour scale represents quality factor as defined in section 4.1, the left plot depicts  $Q$  and the right plot shows  $Q'$ . The arrows indicate the best tuning. The small white disc in the left plot indicates the control accuracy for the etalon tuning, see section 4.3.

### 4.1 Quality function

In order to compare different tunings effectively we will use a quality functions. The quality of a specific etalon setting, given by the tuning of the West and North etalon, can be defined as:

$$Q = \frac{I_{0,0}}{I_{N,W}} \quad (1)$$

with  $I_{N,W}$  being the square of the CARM to DARM transfer function integrated from 50 Hz to 2000 Hz:

$$I_{N,W} = \int_{50}^{2000} (C_{N,W}(f))^2 df \quad (2)$$

Alternatively we can also weigh the transfer function by frequency and define the function as:

$$Q' = \frac{I'_{0,0}}{I'_{N,W}} \quad (3)$$

with

$$I'_{N,W} = \int_{50}^{2000} (C_{N,W}(f))^2 / f df \quad (4)$$




Figure 4: The CARM to DARM transfer functions for the best, worst and nominal etalon tunings.

## 4.2 Matched, plane-wave model (maxtem off)

The effect of the etalon tunings on the CARM to DARM coupling over the full parameter space is shown in figure 3. We can see that both types of quality functions result in an overall similar structure with high quality values in the same region. The exact best tuning differ for the two functions and the structures for the frequency-weighted function  $Q'$  yields slightly sharper structures. Figure 4 shows the full transfer function for the best, worst and nominal (both tunings are zero) tunings, providing a visual guide for the range of change that can be achieved.

## 4.3 Control requirements

In order to benefit from a good tuning value, the etalons must be kept at the correct temperature with an active feedback loop. In this section we plot zoomed in views and cross sections of the plot shown in figure 3 to estimate the requirements for the accuracy of these control loops. The currently achievable accuracy is estimated to be 6 mK RMS.

Figure 5 shows a zoom into the  $Q'$  plot, and figure 6 shows the vertical and horizontal cross section around the optimal point. Several visual guides are shown to understand the impact control loop noise. Figure 7 shows an alternative selection for the cross sections, picking the broadest and narrowest features. The cross sections are shown in figure 8. However, the features in this new cross sections is not significantly narrower than in the horizontal and vertical slices shown before. Finally, figure 9 shows the transfer function for the optimal setting and for two settings with the West etalon being detuned by a temperature change of plus/minus 3 mK. Together these plots show that a modest improvement in control accuracy is required to be able to benefit fully from the optimal etalon tunings.

However it should be noted that the plots in figure 3 show a broad good region near  $N = 90, W = 110$ , which would already give a good improvements without the need for better controls. Figures 10 and 11 show a zoomed in view and cross sections for this region for the  $Q$  quality function.




Figure 5: Zoomed-in view of  $Q'$  in figure 3. The yellow contour line shows where the quality is equal to 7 (roughly half of the peak value of 13.6). The diameter of the white disk represents a 6 mK temperature change in the mirrors. The blue lines show the location of the cross sections plotted in figure 6.




Figure 6: Cross sections for the date shown in figure 5. The ragged look of the data is an artefact of the way in which the simulated data has been generated.




Figure 7: Zoomed-in view of  $Q'$  in figure 3, indicating the location of the cross sections shown in figure 8.




Figure 8: Cross sections for the date shown in figure 7.




Figure 9: Several CARM to DARM transfer functions related to figure 5. One trace refers to the optimal point, two more traces show the transfer function if the West etalon is detuned by plus/minus 3 mK. (The fourth trace shows an arbitrary example transfer function for the point at the origin of figure 5.)




Figure 10: Zoomed-in view of  $Q$  in figure 3, indicating the location of the cross sections shown in figure 11. Here the broader central region was chosen where less control accuracy would be required. The yellow trace shows where the quality has dropped to approximately half (4 for a maximum of 8,7).




Figure 11: Cross sections for the date shown in figure 10.

#### 4.4 Mis-matched model (maxtem 2)

As a final check, the CARM to DARM coupling was re-computed for a more realistic model, in which the arm cavities are not perfectly matched. The target was a  $\approx 1\%$  to  $1.5\%$  mismatch of the arm cavities. In the simulation this achieved by changing the curvature of the NE and WE end mirrors:

$$R_{\text{NE}} = 1695 \rightarrow 1655 \quad (5)$$

$$R_{\text{WE}} = 1696 \rightarrow 1658 \quad (6)$$

This produced the following mode mismatch coefficients, which show mismatches between 1.2% and 1.6% between the arm cavities and the power recycling cavity respectively:

	cavOMC1	cavW	cavN	cavPRW	cavPRN	cavSRW	cavSRN
cavOMC1	0.000000	0.007707	0.009477	0.000759	0.000599	0.000043	0.000014
cavW	0.007707	0.000000	0.000093	0.013224	0.012532	0.008845	0.008281
cavN	0.009477	0.000093	0.000000	0.015500	0.014752	0.010733	0.010111
cavPRW	0.000759	0.013224	0.015500	0.000000	0.000010	0.000448	0.000588
cavPRN	0.000599	0.012532	0.014752	0.000010	0.000000	0.000327	0.000448
cavSRW	0.000043	0.008845	0.010733	0.000448	0.000327	0.000000	0.000009
cavSRN	0.000014	0.008281	0.010111	0.000588	0.000448	0.000009	0.000000

The resulting transfer function as shown in figure 12 show a prominent  $1/f$  sloped coupling whose amplitude changes with the etalon tuning, similar behaviour to the matched model. However, one can see also a large frequency independent component that changes with the etalon tuning. That behavior is similar to what experimentally observed during O3, which would indicate that mis-match could play a significant role in the frequency noise coupling.




Figure 12: Etalon with higher-order modes and a small mode-mismatch (maxtem 2)

## References

- [1] S. Hild, A. Freise, M. Mantovani, S. Chelkowski, J. Degallaix and R. Schilling: ‘Using the etalon effect for in situ balancing of the Advanced Virgo arm cavities’ *Classical and Quantum Gravity* **26** 025005 (12pp) 2 (2009) <http://stacks.iop.org/0264-9381/26/025005> 1
- [2] A. Freise: source code and input files for the simulations shown in this note are available in the LIGO git-lab repository [https://git.ligo.org/IF0sim/Finesse\\_playground/-/tree/master/andreas\\_freise/2020/adv\\_etalon](https://git.ligo.org/IF0sim/Finesse_playground/-/tree/master/andreas_freise/2020/adv_etalon) 2

- [3] A. Allocca and M. Mantovani: ‘AdV etalon control’, Virgo presentation VIR-0890A-19 (2019) <https://tds.virgo-gw.eu/ql/?c=14724> 1
- [4] A. Allocca: ‘Etalon for AdV+ phase II - kick start discussion’ Virgo presentation VIR-0266A-20 (2020) <https://tds.virgo-gw.eu/ql/?c=15387> 1
- [5] D. Brown and A. Freise: ‘Finesse’ (2014) <http://www.gwoptics.org/finesse> 1
- [6] D. D. Brown, P. Jones, S. Rowlinson, S. Leavey, A. C. Green, D. Töyrä and A. Freise: ‘Pykat: Python package for modelling precision optical interferometers’ *SoftwareX* **12** 100613 (2020) <http://www.sciencedirect.com/science/article/pii/S2352711020303265> 1
- [7] A. Chiummo and G. Vajente: ‘ITF asymmetries and coupling of laser technical noises’ Virgo technical note VIR-0043A-12 (2012) <https://tds.virgo-gw.eu/ql/?c=8845> 2
- [8] J. Brooks, M. Mantovani, A. Allocca, J. Casanueva Diaz, V. Dattilo, A. Masserot and P. Ruggi: ‘Temperature Control for an Intra-Mirror Etalon in Interferometric Gravitational Wave Detector Fabry–Perot Cavities’ *Galaxies* **8** 80 (2020) <https://www.mdpi.com/2075-4434/8/4/80>

## 5 Appendix 1: the Finesse file

```
%%% FTblock header
#-----
# An Advanced Virgo plus input file for Finesse
# file based on:
# https://wiki.virgo-gw.eu/Commissioning/OptSimulation/FinesseFile
# by Michele Valentini, copied August 2020
# Original comments:
# - dual recycled Adv + interferometer, with 40% SRM transmission and 40w input power.
# - No squeezing is considered yet in this file.
# - OMCs are added as defined in the (old) Adv TDR, if new OMCs will be
# installed in Adv +, the file needs to be updated;
# - Optics parameters are set to measured (if available) or Adv TDR COLD ifo values.
# No thermal effects or ring heater effects are considered;
# - PRM and SRM RoCs are both set to 1430 m in order to make the SRCL and PRCL cavities
# "compatible" with each other;
# Compensation plate focals are arbitrarily (but realistically) set in order to optimize
# the matching of the recycling cavities with the arms and to keep the recycling cavitites
# stable but very close to marginal stability (2mrad RT Gouy phase shft);
# - Modulators at f1 (6.27 MHz), 5*f1, 7*f1 have been added in order to test particular error
# signals. They can be disabled easily by putting ther modulation index to 0;
# - Superattenuators mechanical transfer function is missing, therefore radiation pressure
# effect is not simulated correctly unless they are properly added;

# Further changes by Andreas Freise:
# - moved RC/M attr commands to the components (better for block removal)
# - removed extra modulators
# - clarify conventions X = N , Y = W
# - removed tunings const/block, as this is now handled with Pykat
# - changed to single OMC for O4 configuration
# - matching the OMC, Michele Valentini: better match after
# changing the focal lenses of MMT_L1 and MMT_L2 to 8.589e+00 and 1.105e+00
# - adding 329 ppm reflectivity to main BS AR surface
# - renamed space 'dummy' to allow lock drag

%%% FTend header

%%% FTblock Laser
#####
l i1 40 0 0 nin
s s0 1m nin nEOM1a
lambda 1.064u

#####
%%% FTend Laser

%%% FTblock EOMs
#####

# modulation frequencies [TDR, tab 2.3, pp 24]
const f1 6270777          # fmod1 in TDR
```

```
const f2 56436993          # 9 * f1, fmod2 in TDR
const f3 8361036          # 4 / 3 * f1, fmod3 in TDR

# EOMs with modulation indices from https://logbook.virgo-gw.eu/virgo/?r=34898
# and https://logbook.virgo-gw.eu/virgo/?r=38123
# and https://logbook.virgo-gw.eu/virgo/?r=41551
mod EOM1 $f1 0.22 1 pm 0 nEOM1a nEOM1b
s sEOM1 0.1 nEOM1b nEOM2a
mod EOM2 $f2 0.16 1 pm 0 nEOM2a nEOM2b
s sEOM2 0.1 nEOM2b nEOM3a
mod EOM3 $f3 0.15 1 pm 0 nEOM3a nEOM3b
s s1 0.2 nEOM3b nFI1
#####
%% FTend EOMs

%% FTblock REFL
#####

# Lossless faraday isolator to pick off B2-beam
dbs FI nFI1 nFI2 nFI3 nB2a
s sdummy 1 nFI2 nD
# Beam splitter to obtain 18 mW of power on B2 (with 40W input)
s sFI_B2_a 0 nB2a nB2att
bs B2_attenuator 0.99816 0.00184 0 0 nB2att dump nB2 dump

# Space between the isolator and PR
s sFI_PR 0 nFI3 nPR1

#####
%% FTend REFL

%% FTblock PRC
#####

# Power recycling mirror. Measured thickness (ref?). The AR-surface is
# wedged, therefore the AR-reflectivity is set as a loss.
m2 PRAR 0 $L_PRAR 0 nPR1 nPRsub1
s sPRsub 0.1003 $nsilica nPRsub1 nPRsub2
m1 PR $T_PR $L_PR 0 nPRsub2 nPR2
#attr PR Rc -1477          # Measured cold IFO PR RoC [VIR-0029A-15]
attr PR Rc -1430          # Design value to have good matching (should correspond to warm compensated ifo)
attr PRAR Rc -3.62        # Measured PR AR RoC [VIR-0029A-15]

# Space between PR and POP. Length from TDR.
s lPR_POP 0.06 1 nPR2 nPOP1

# Pick off plate. The angle of incidence and the physical distance the beam
# propagates inside POP are computed from thickness of 3.5 cm [TDR], 6 deg
# tilt [TDR], and refractive index of $nsilica. POP AR is wedged, thus,
# the AR-reflectivity is set as a loss.
bs2 POP_AR 0 $L_POP2 0 6.0 nPOP1 nPOPunused1 nPOPsub1 nPOPsub3
s sPOPsub 0.03549 $nsilica nPOPsub1 nPOPsub2
bs2 POP $R_POP1 0 0 4.135015 nPOPsub2 nPOPsub4 nPOP2 nB4
```

```
s sB4_att 0 nB4 nB4att
bs B4_attenuator 0.7344 0.2656 0 0 nB4att dump nB4b dump

# Space between POP and BS. Measured. Reference?
s lPOP_BS 5.9399 nPOP2 nBSs

#####
%% FTend PRC

%% FTblock BS
#####
# The beam splitter

# Angles of incidence and the physical distance the beam propagates inside
# the BS substrate are computed using BS thickness of 6.5 cm, angle of
# incidence of 45 degrees, and refractive index of nsilica. All from TDR.

# HR surface
bs1 BS $T_BS $L_BS 0 -45 nBSs nBSw nBSsub1a nBSsub2a
attr BS M 34

# Substrate
s sBSsub1 0.074459 $nsilica nBSsub1a nBSsub1b
s sBSsub2 0.074459 $nsilica nBSsub2a nBSsub2b
# AR-surface towards north cavity
bs2 BSAR1 329u $L_BSAR 0 -29.1951 nBSsub1b nUnused1 nBSn nBSAR
# AR-surface towards the dark port
bs2 BSAR2 329u $L_BSAR 0 -29.1951 nBSsub2b nUnused2 nBSe nUnused3

#####
%% FTend BS

%% FTblock Narm
#####

# Distance between beam splitter and compensation plate. Measured. Ref?
s lBS_CPN 5.3662 nBSn nCPN1

# Compensation plate. Thickness from [TDR, tab 2.9]
m2 CPN1 0 $L_CPN1 0 nCPN1 nCPNsub1
s sCPNsub 0.035 $nsilica nCPNsub1 nCPNsub2
m2 CPN2 0 $L_CPN2 0 nCPNsub2 nCPN2

# Space between compensation plate and thermal lens.
s sCPN_TL 0 nCPN2 nCPN_TL1

# Thermal lens in compensation plate
lens CPN_TL $f_CPN_TL nCPN_TL1 nCPN_TL2

# Space between compensation plate and NI. From TDR.
s sCPN_NI 0.2 nCPN_TL2 nNI1

# North input mirror. The AR-surface is not wedged, thus the
```

```
# AR-reflectivity is set as a reflectivity. Measured thickness. Ref?
m2 NIAR $R_NIAR $L_NIAR 0 nNI1 nNIsub1
s sNIsub 0.20026 $nsilica nNIsub1 nNIsub2
m1 NI $T_NI $L_WI 0 nNIsub2 nNI2
attr NI Rc -1424.6      # Measured cold IFO NI RoC [VIR-0544A-14]
attr NI M 42
# attr NIAR Rc -1420    # Design NI AR RoC [TDR, table 2.6]
attr NIAR Rc -1424.6    # Approximately the same as measured HR surface
```

```
# Space between north test masses.
s LN 2999.8 nNI2 nNE1
```

```
# North end mirror. The AR-surface is wedged, thus the
# AR-reflectivity is set as a loss. Thickness from TDR
m1 NE $T_NE $L_NE 0 nNE1 nNEsub1
s sNEsub 0.2 $nsilica nNEsub1 nNEsub2
m2 NEAR 0 $L_NEAR 0 nNEsub2 nNE2
attr NE Rc 1695        # Measured cold IFO NE RoC [VIR-0269A-15]
attr NE M 42
```

```
#####
%% FTend Narm
```

```
%% FTblock Warm
```

```
#####
```

```
# BS to compensation plate. Measured. Ref?
s LBS_CPW 5.244 nBSw nCPW1
```

```
# Compensation plate CP02. Thickness from [TDR, tab 2.9]
m2 CPW1 0 $L_CPW1 0 nCPW1 nCPWsub1
s sCPWsub 0.035 $nsilica nCPWsub1 nCPWsub2
m2 CPW2 0 $L_CPW2 0 nCPWsub2 nCPW2
```

```
# Space between compensation plate and the thermal lens
s sCPW_TL 0 nCPW2 nCPW_TL1
```

```
# Thermal lens in compensation plate
lens CPW_TL $f_CPW_TL nCPW_TL1 nCPW_TL2
```

```
# Space between compensation plate and WI. From TDR.
s sCPW_WI 0.2 nCPW_TL2 nWI1
```

```
# West input mirror. The AR-surface is not wedged, thus the
# AR-reflectivity is set as a reflectivity.
m2 WIAR $R_NIAR $L_NIAR 0 nWI1 nWIsub1
s sWIsub 0.20031 $nsilica nWIsub1 nWIsub2
m1 WI $T_WI $L_WI 0 nWIsub2 nWI2
attr WI Rc -1424.5     # Measured cold IFO WI RoC [VIR-0543A-14]
attr WI M 42
# attr WIAR Rc -1420   # Design WI AR RoC [TDR, table 2.6]
attr WIAR Rc -1424.5   # Approximately the same as measured HR surface
```

```

# Space between west test masses
s LW 2999.8 nWI2 nWE1

# West end mirror. The AR-surface is wedged, thus the
# AR-reflectivity is set as a loss.
m1 WE $T_WE $L_WE 0 nWE1 nWesub1
s sWesub 0.2 $nsilica nWesub1 nWesub2
m2 WEAR 0 $L_WEAR 0 nWesub2 nWE2
attr WE Rc 1696          # Measured cold IFO WE RoC [VIR-0270A-15]
attr WE M 42

#####
%% FTend Warm

%% FTblock SRC
#####
# TODO get length reference
s lsr 5.943 nBSe nSR1

m1 SR $T_SR $L_SR 0 nSR1 nSRsub1
s sSRsub 0.1004 $nsilica nSRsub1 nSRsub2  ## VIR-0028A-15
m2 SRAR 0 $L_SRAR 0 nSRsub2 nSR2
#attr SR Rc 1443          # Measured cold IFO SR RoC [VIR-0028A-15]
attr SR Rc 1430          # Changed to match the PRC (should correspond to warm compensated ifo)
attr SRAR Rc 3.59       # Design [TDR, table 2.8]

#####
%% FTend SRC

%% FTblock OMCpath
#####
# All parameters in the block are from the TDR, table 7.16.

s sSR_MMTL 4.451 nSR2 nMMT_La

# Meniscus lens. Focal length obtained via lensmaker's equation with
# thin lens approximation, and assuming n = 1.44963.
lens MMT_L -3.596 nMMT_La nMMT_Lb

s sMMT_ML_M1 0.6 nMMT_Lb nMMT_M1a

# Telescope mirror 1
bs MMT_M1 1 0 0 0 nMMT_M1a nMMT_M1b nMMT_M1c nMMT_M1d
attr MMT_M1 Rc 1.44

s sMMT_M1_M2 0.765 nMMT_M1b nMMT_M2a

# Telescope mirror 2
bs MMT_M2 1 0 0 0 nMMT_M2a nMMT_M2b nMMT_M2c nMMT_M2d
attr MMT_M2 Rc 0.09

s sMMT_M2_L1 0.5 nMMT_M2b nMMT_L1a

```

```
# Modematching lenses.
# -----
# Focal length mnually adjusted for a match to OMC, should
# be replaced by proper beam path

lens MMT_L1 8.589 nMMT_L1a nMMT_L1b
s sMMT_L1_L2 0.12 nMMT_L1b nMMT_L2a
lens MMT_L2 1.105 nMMT_L2a nMMT_L2b

#####
%% FTend OMCpath

%% FTblock OMC
#####
# OMC parameters are preliminary

s sL2_OMC1 0.99 nMMT_L2b nOMC1_1

# OMC
# -----
bs OMC1_1 0.9969 0.0031 0 8.876 nOMC1_1 nOMC1_refl nOMC1_2 nOMC1_3
s sOMC1_1 0.0600 $nsilica nOMC1_2 nOMC1_4
bs OMC1_2 0.9969 0.0031 0 8.876 nOMC1_4 nOMC1_5 nOMC1_out dump
s sOMC1_2 0.0630 $nsilica nOMC1_5 nOMC1_6
bs OMC1_3 0.9999665 33.5u 0 8.876 nOMC1_6 nOMC1_7 dump dump
s sOMC1_3 0.0600 $nsilica nOMC1_7 nOMC1_8
bs OMC1_4 0.9999665 33.5u 0 8.876 nOMC1_8 nOMC1_9 dump dump
s sOMC1_4 0.0630 $nsilica nOMC1_9 nOMC1_3

# Design OMC RoCs [TDR page 247]
attr OMC1_3 Rc 1.499
attr OMC1_4 Rc 1.499
cav cavOMC1 OMC1_1 nOMC1_2 OMC1_1 nOMC1_3
#####
%% FTend OMC

# Space to output B1
s sOut 1 nOMC1_out nB1

%% FTblock Gaussian
#####
# Arms
cav cavW WI nWI2 WE nWE1
cav cavN NI nNI2 NE nNE1
# PRC reversed, works better with mismatch_cavties
cav cavPRW WI nWIsup2 PR nPR2
cav cavPRN NI nNIsup2 PR nPR2
#cav cavPRW PR nPR2 WI nWIsup2
#cav cavPRN PR nPR2 NI nNIsup2
# SRC
cav cavSRW SR nSR1 WI nWIsup2
cav cavSRN SR nSR1 NI nNIsup2
```

```
#####  
%% FTend Gaussian  
  
%% FTblock RTLs  
#####  
# Specifying reflectivities, transmissions, and losses.  
  
# Transmissions. All measured. These are set to HR-surfaces.  
# -----  
const T_NI 0.01377      # NI transmission [IM04, VIR-0544A-14]  
const T_WI 0.01375      # WI transmission [IM02, VIR-0543A-14]  
const T_WE 4.3u         # WE transmission [EM03, VIR-0270A-15]  
const T_NE 4.4u         # NE transmission [EM01, VIR-0269A-15]  
const T_BS 0.5012       # BS transmission [VIR-0446B-14]  
const T_PR 0.04835      # PR transmission [VIR-0029A-15]  
const T_SR 0.4  
# SR transmissivity optimized for 04 https://wiki.virgo-gw.eu/AdvancedVirgoPlus/Meeting190503  
  
# Losses  
# -----  
# Set to match measured roundtrip loss of 61 +- 5 ppm [Loggbook 38601]  
const L_NI 27u          # NI loss  
const L_NE 27u          # NE loss  
# Set to match measured roundtrip loss of 56 +- 5 ppm [Loggbook 38601]  
const L_WI 27u          # WI loss  
const L_WE 27u          # WE loss  
# These are included in the above values.  
# const L_NI 0.19u      # NI Absorbtion coating losses. From Valeria. Ref?  
# const L_WI 0.28u      # WI Absorbtion coating losses. From Valeria. Ref?  
# const L_NE 0.24u      # NE Absorbtion coating losses. From Valeria. Ref?  
# const L_WE 0.24u      # WE Absorbtion coating losses. From Valeria. Ref?  
# Gussed/assumed values  
const L_PR 30u  
const L_BS 30u  
const L_SR 30u  
  
# AR-surfaces  
# -----  
# Measured non-wedged AR-surface reflectivities. Setting as reflectivities  
# TODO remove etalon by default, add via pykat?  
const R_WIAR 58u        # WI AR reflectivity [IM02, VIR-0543A-14]  
const R_NIAR 32u        # NI AR reflectivity [IM04, VIR-0544A-14]  
# Mesured wedged AR-reflectivities. Setting as losses due to wedge  
const L_WEAR 155u       # WE AR reflectivity [EM03, VIR-0270A-15]  
const L_NEAR 133u       # NE AR reflectivity [EM01, VIR-0269A-15]  
const L_BSAR 329u       # BS AR reflectivity [VIR-0446B-14]  
const L_PRAR 160u       # PR AR reflectivity [VIR-0029A-15]  
const L_SRAR 141u       # SR AR (surface 2) reflectivity [VIR-0028A-15]  
  
# Losses  
const L_NIAR 0          # Unknown  
const L_WIAR 0          # Unknown
```

```
# POP and CPs
# -----
# Measured POP-reflectivities. Setting S2 surface as loss due to wedge.
const R_POP1 184u      # POP S1 reflectivity [VIR-0027A-15]
const L_POP2 125u      # POP S2 reflectivity [VIR-0027A-15]

# Masured CP-reflectivities. Set as losses, but not sure if they have wedge
const L_CPW1 87u       # CPW (CP02) surface 1 reflectiviy [VIR-0506B-14]
const L_CPW2 114u      # CPW (CP02) surface 2 reflectiviy [VIR-0506B-14]
const L_CPN1 22u       # CPW (CP03) surface 1 reflectiviy [VIR-0153A-16]
const L_CPN2 44u       # CPW (CP03) surface 2 reflectiviy [VIR-0153A-16]

#####
%% FTend RTLs

%% FTblock Constants
#####
const nsilica 1.44963 # Refractive index of silica
const l_arm 2999.8    # Arm length

# Lenses
# -----
# Compensation plate focal lengths for cold IFO. Values optimised to yield
# a well matched cold interferomter at 40 W with PRM and SRM HR RoCs of 1430m
const f_CPN_TL -3.397e+05 # North
const f_CPW_TL -3.551e+05 # West

# Compensation plate lensing. Values from Valeria.
# const f_CPN_TL 54000 # North
# const f_CPW_TL 930000 # West

#####
%% FTend Constants
```